

## Paper 383-2011

**PATH and LISMOD Languages in SAS<sup>®</sup> PROC TCALIS for Multigroup Comparisons**

Fei Gu, University of Kansas, Lawrence, KS

Wei Wu, University of Kansas, Lawrence, KS

**ABSTRACT**

This article introduced a new SAS procedure, PROC TCALIS, in structural equation modeling (SEM). In the previous version of SAS, PROC CALIS was incapable of handling multigroup models, whereas in SAS 9.2, PROC TCALIS, is enhanced with many new features such as capability for multi-group analysis and for accommodating different model specification languages used in other major SEM software packages (e.g., EQS, Mplus and LISREL). Two types of modeling languages are illustrated in a multigroup measurement model with and without equality constraints across groups. Fit indices and parameter estimates from SAS are compared to those from Mplus and LISREL. The fit indices from SAS are essentially consistent with those from Mplus but not those from LISREL. The parameter estimates are identical across the three software packages.

**INTRODUCTION**

In past decades, structural equation modeling (SEM) has gained increasing popularity in a wide variety of research situations. Many specialized software packages have been developed for SEM, including EQS (Bentler & Wu, 2002), Mplus (Muthén & Muthén, 1998-2007), LISREL (Jöreskog & Sörbom, 1996), Mx (Neale, Boker, Xie, & Maes, 2003), and AMOS (Arbuckle, 2003). Since version 8, SAS has also added a procedure, PROC CALIS, to accommodate SEM. Compared to other specialized SEM software, however, PROC CALIS has one major limitation in SEM: its inability to implement multigroup comparisons (e.g., Fan & Fan, 2005). The fact that some researchers tried to "trick" SAS to analyze multigroup models (provided that each group had the same sample size), though instructive, is not applicable to general situations where group sizes are mostly different (Marcoulides & Hershberger, 1997; Jones-Farmer, Pitts, & Rainer, 2008). In addition, using such a trick may give an incorrect degree of freedoms. Thus, one must be cautious about using PROC CALIS for multigroup analyses. Consequently, SAS is not the first choice to implement multigroup invariance tests (Jones-Farmer et al., 2008). According to Byrne (2004), most literature addressing multigroup invariance has used either LISREL or EQS.

Although there are many other packages adept at handling multigroup analysis, there might be times when, by convenience or necessity, SAS must be used. As a versatile and comprehensive system for data management, programming, and statistical analyses, SAS offers quantitative researchers an extremely flexible environment to conduct various Monte Carlo simulation studies (Fan, Felsövályi, Sivo, & Keenan, 2003). SEM simulation work using PROC CALIS existed in previous years and can be found in the literature (e.g., Fan & Sivo, 2005; Yang & Green, 2010).

In SAS 9.2, PROC TCALIS is modified from PROC CALIS with changes and enhancements. According to the SAS document (SAS Institute, 2008), PROC TCALIS is not a simple functional enhancement of PROC CALIS. The basic computational architecture of PROC TCALIS is quite different from that of PROC CALIS. New features include, but are not limited to, new modeling languages, multigroup analysis, and improved mean structures analysis.

The new modeling languages reflect different modeling terminology and philosophies. Traditionally, the LINEQS language in PROC CALIS, which is similar to EQS (Bentler & Wu, 2002), was frequently used to specify SEM models (e.g., Brown, 2006). In PROC TCALIS, the flexible PATH language that is similar to the language used in Mplus (Muthén & Muthén, 1998-2007) makes researchers almost ready to translate any path diagram into the PATH model. A model specified by using the PATH language is referred to as a PATH model for simplicity. In addition, the LISMOD language, which stands for LISREL MODELing, caters to the need of users who are used to LISREL models (Jöreskog & Sörbom, 1996). Therefore, users of other major SEM software, e.g., EQS, Mplus and LISREL, may find it easy to use PROC TCALIS by choosing one of the modeling languages that is most convenient and familiar to them.

The purpose of this article is to illustrate the use of the two new modeling languages in PROC TCALIS, the PATH and LISMOD languages, for multigroup analysis by using the same data presented by Jones-Farmer et al. (p. 165, 2008). To show the validity of the results from PROC TCALIS, the results are also compared to those obtained from Mplus 5.21 and LISREL 8.80.

## EXAMPLE

Details of the data and the measurement model were described by Jones-Farmer et al. (2008) below:

"The data for this example were collected in a public safety organization undergoing an organizational merger. The new organization is the result of the consolidation of previously autonomous organizations providing fire suppression and emergency medical services in a midsized southeastern U.S. city. The variables considered in this example were selected from the results of a larger self-report quantitative survey. The survey data were collected on two separate groups with samples of size  $n_1=213$  and  $n_2=213$ . Details regarding the study are available in Pitts (2006).

It is necessary to establish the invariance of three constructs across the two groups. The first construct relates to a specific dimension from the Communication Satisfaction Questionnaire (CSQ; Downs & Hazen, 1977) that relates to an employee's satisfaction with the communication between themselves and their immediate supervisor or manager. This construct, referred to in the analysis as F\_comm, consists of six items. The second and third constructs result from two justice dimensions from scales developed by Niehoff and Moorman (1993). The Procedural Justice scale consists of six items that assess an employee's perception of the fairness of the process and procedures by which decisions regarding the merger are made. The Distributive Justice scale, also a six-item scale, assesses an employee's perception of the fairness of work outcomes related to the merger such as pay, workload, job responsibilities, and recognition. Procedural and distributive justice are referred to as F\_pj and F\_dj, respectively, in the analysis (p.156)."

The path diagram depicting the measurement model is shown in Figure 1 (see also Jones-Farmer et al., 2008, p.157). Steps in this illustration strictly follow those of Jones-Farmer et al. (2008). First, a baseline model is fit to the data for each group separately. Second, an unconstrained model is estimated for both groups simultaneously. Third, cross-group constraints are placed on the corresponding factor loadings to specify a constrained model.




Figure 1. Path diagram of the measurement model.

The SAS program, using PROC TCALIS, that is used in this example is given in Table 1, with line numbers given for reference. Lines 1 through 47 of Table 1 input two covariance matrices for Groups 1 and 2 separately.

**Table 1.** Sample SAS Program for Example

```

1 data group1(type=cov);
2   _type_="COV";
3   input _NAME_ $ com1-com6 disj1-disj6 proj1-proj6;
4   cards;
5   com1  4.8 4.2 4.1 3.8 3.6 3.2 0.7 0.7 0.7 0.7 1.0 0.4 0.7 1.2 1.1 0.9 1.2 0.4
6   com2  4.2 4.9 4.0 4.1 3.5 3.3 0.8 0.8 0.9 0.8 1.1 0.5 0.8 1.2 1.2 1.1 1.3 0.7
7   com3  4.1 4.0 4.8 3.9 3.6 3.3 0.8 0.8 0.8 0.6 0.9 0.4 0.7 1.2 1.1 1.2 1.3 0.8
8   com4  3.8 4.1 3.9 4.7 3.5 3.2 0.9 1.0 0.9 0.8 1.3 0.7 0.9 1.2 1.3 1.2 1.2 0.6
9   com5  3.6 3.5 3.6 3.5 4.5 3.5 0.6 0.8 0.7 0.6 0.9 0.4 0.8 0.9 1.0 0.9 1.0 0.5
10  com6  3.2 3.3 3.3 3.2 3.5 4.8 0.5 0.5 0.4 0.4 0.6 0.1 0.9 0.9 0.9 0.9 0.9 0.3
11  disj1 0.7 0.8 0.8 0.9 0.6 0.5 3.7 3.1 3.1 2.9 2.6 1.8 1.4 1.3 1.4 1.2 1.1 1.0
12  disj2 0.7 0.8 0.8 1.0 0.8 0.5 3.1 3.4 3.0 3.0 2.7 1.8 1.4 1.1 1.2 0.9 0.9 0.8
13  disj3 0.7 0.9 0.8 0.9 0.7 0.4 3.1 3.0 3.6 3.1 2.7 1.6 1.5 1.2 1.2 1.1 1.0 0.8
14  disj4 0.7 0.8 0.6 0.8 0.6 0.4 2.9 3.0 3.1 3.3 2.7 1.6 1.5 1.2 1.2 1.1 1.0 0.7
15  disj5 1.0 1.1 0.9 1.3 0.9 0.6 2.6 2.7 2.7 2.7 3.2 1.7 1.5 1.4 1.3 1.1 1.2 0.7
16  disj6 0.4 0.5 0.4 0.7 0.4 0.1 1.8 1.8 1.6 1.6 1.7 3.4 1.2 0.8 0.9 0.7 0.7 0.5
17  proj1 0.7 0.8 0.7 0.9 0.8 0.9 1.4 1.4 1.5 1.5 1.5 1.2 3.4 1.4 1.4 1.1 1.3 0.6
18  proj2 1.2 1.2 1.2 1.2 0.9 0.9 1.3 1.1 1.2 1.2 1.4 0.8 1.4 2.5 2.0 1.4 1.6 1.2
19  proj3 1.1 1.2 1.1 1.3 1.0 0.9 1.4 1.2 1.2 1.2 1.3 0.9 1.4 2.0 2.4 1.6 1.7 1.1
20  proj4 0.9 1.1 1.2 1.2 0.9 0.9 1.2 0.9 1.1 1.1 1.1 0.7 1.1 1.4 1.6 2.6 1.6 1.3
21  proj5 1.2 1.3 1.3 1.2 1.0 0.9 1.1 0.9 1.0 1.0 1.2 0.7 1.3 1.6 1.7 1.6 2.5 1.2
22  proj6 0.4 0.7 0.8 0.6 0.5 0.3 1.0 0.8 0.8 0.7 0.7 0.5 0.6 1.2 1.1 1.3 1.2 3.1
23 ;
24 data group2(type=cov);
25   _type_="COV";
26   input _NAME_ $ com1-com6 disj1-disj6 proj1-proj6;
27   cards;
28   com1  4.2 3.2 2.7 2.8 2.4 2.4 1.0 0.9 0.9 0.9 0.7 0.9 0.8 0.6 0.9 0.6 0.8 1.0
29   com2  3.2 3.8 2.3 2.7 2.5 2.4 0.8 0.9 0.8 0.8 0.7 1.1 0.7 0.5 0.9 0.7 0.9 0.9
30   com3  2.7 2.3 3.9 1.9 1.6 1.4 0.5 0.5 0.6 0.3 0.3 0.4 0.4 0.2 0.4 0.5 0.5 0.6
31   com4  2.8 2.7 1.9 4.2 2.3 2.2 0.8 0.8 0.8 0.9 0.6 1.0 0.9 0.8 0.9 0.7 0.9 0.9
32   com5  2.4 2.5 1.6 2.3 3.7 2.3 0.5 0.4 0.3 0.3 0.4 0.5 0.4 0.4 0.5 0.2 0.6 0.5
33   com6  2.4 2.4 1.4 2.2 2.3 3.7 1.1 0.7 0.9 0.9 0.7 1.2 0.3 0.3 0.7 0.5 0.7 0.7
34   disj1 1.0 0.8 0.5 0.8 0.5 1.1 3.1 2.0 2.4 2.3 1.7 1.4 0.9 1.0 0.7 1.2 0.9 0.8
35   disj2 0.9 0.9 0.5 0.8 0.4 0.7 2.0 3.0 2.1 2.2 2.0 1.5 1.2 1.3 1.1 1.0 1.2 0.7
36   disj3 0.9 0.8 0.6 0.8 0.3 0.9 2.4 2.1 2.9 2.7 2.0 1.6 1.2 1.3 1.0 1.2 1.3 0.9
37   disj4 0.9 0.8 0.3 0.9 0.3 0.9 2.3 2.2 2.7 3.0 2.1 1.7 1.1 1.2 1.1 1.2 1.3 0.9
38   disj5 0.7 0.7 0.3 0.6 0.4 0.7 1.7 2.0 2.0 2.1 2.5 1.5 1.0 1.3 1.1 0.8 1.0 0.7
39   disj6 0.9 1.1 0.4 1.0 0.5 1.2 1.4 1.5 1.6 1.7 1.5 3.6 0.7 1.1 1.2 0.9 1.2 1.0
40   proj1 0.8 0.7 0.4 0.9 0.4 0.3 0.9 1.2 1.2 1.1 1.0 0.7 3.1 1.6 1.1 1.0 1.5 1.0
41   proj2 0.6 0.5 0.2 0.8 0.4 0.3 1.0 1.3 1.3 1.2 1.3 1.1 1.6 2.3 1.4 1.0 1.5 1.0
42   proj3 0.9 0.9 0.4 0.9 0.5 0.7 0.7 1.1 1.0 1.1 1.1 1.2 1.1 1.4 2.5 1.3 1.4 0.8
43   proj4 0.6 0.7 0.5 0.7 0.2 0.5 1.2 1.0 1.2 1.2 0.8 0.9 1.0 1.0 1.3 2.5 1.2 1.0
44   proj5 0.8 0.9 0.5 0.9 0.6 0.7 0.9 1.2 1.3 1.3 1.0 1.2 1.5 1.5 1.4 1.2 2.5 1.1
45   proj6 1.0 0.9 0.6 0.9 0.5 0.7 0.8 0.7 0.9 0.9 0.7 1.0 1.0 1.0 0.8 1.0 1.1 3.2
46 ;
47 run;
48
49 ***** using PATH language *****;
50 %macro BasePath;
51 path
52   com1  <- F_comm  1.0,    /** F_comm -> com1  1.0,  ***/
53   com2  <- F_comm  c2,     /** F_comm -> com2  c2   ***/
54   com3  <- F_comm  c3,     /** F_comm -> com3  c3   ***/
55   com4  <- F_comm  c4,     /** F_comm -> com4  c4   ***/
56   com5  <- F_comm  c5,     /** F_comm -> com5  c5   ***/
57   com6  <- F_comm  c6,     /** F_comm -> com6  c6   ***/
58   disj1 <- F_dj    1.0,    /** F_dj   -> disj1 1.0,  ***/
59   disj2 <- F_dj    d2,     /** F_dj   -> disj2 d2,   ***/

```

```

60   disj3 <- F_dj      d3,      /***   F_dj    -> disj2 d3,    ***/
61   disj4 <- F_dj      d4,      /***   F_dj    -> disj2 d4,    ***/
62   disj5 <- F_dj      d5,      /***   F_dj    -> disj2 d5,    ***/
63   disj6 <- F_dj      d6,      /***   F_dj    -> disj2 d6,    ***/
64   proj1 <- F_pj      1.0,     /***   F_pj    -> proj1 1.0,   ***/
65   proj2 <- F_pj      p2,      /***   F_pj    -> proj2 p2,    ***/
66   proj3 <- F_pj      p3,      /***   F_pj    -> proj2 p3,    ***/
67   proj4 <- F_pj      p4,      /***   F_pj    -> proj2 p4,    ***/
68   proj5 <- F_pj      p5,      /***   F_pj    -> proj2 p5,    ***/
69   proj6 <- F_pj      p6;      /***   F_pj    -> proj2 p6;    ***/
70
71   pvar
72     F_comm F_dj F_pj=VF__,
73     com1   com2   com3   com4   com5   com6 =vec__,
74     disj1  disj2  disj3  disj4  disj5  disj6=ved__,
75     proj1  proj2  proj3  proj4  proj5  proj6=vep__;
76
77   pcov
78     F_comm F_dj=psi12,
79     F_dj F_pj=psi23,
80     F_comm F_pj=psi13;
81   %mend;
82
83   /* TCALIS - 1st group */
84   proc tcalis data=group1 nobs=213;
85     %BasePath
86   run;
87
88   /* TCALIS - 2nd group */
89   proc tcalis data=group2 nobs=213;
90     %BasePath
91   run;
92
93   /* multigroup: unconstrained model */
94   proc tcalis;
95     group 1 / data=group1 nobs=213;
96     group 2 / data=group2 nobs=213;
97     model 1 / group=1;
98     %BasePath
99     model 2 / group=2;
100    refmodel 1 / AllNewParms;
101   run;
102
103   /* multigroup: constrained model */
104   proc tcalis;
105     group 1 / data=group1 nobs=213;
106     group 2 / data=group2 nobs=213;
107     model 1 / group=1;
108     %BasePath
109     model 2 / group=2;
110     refmodel 1;
111     pvar
112       F_comm F_dj F_pj=G2_VF__,
113       com1   com2   com3   com4   com5   com6 =G2_vec__,
114       disj1  disj2  disj3  disj4  disj5  disj6=G2_ved__,
115       proj1  proj2  proj3  proj4  proj5  proj6=G2_vep__;
116
117     pcov
118       F_comm F_dj=G2_psi12,
119       F_dj F_pj=G2_psi23,
120       F_comm F_pj=G2_psi13;
121   run;
122

```

```

123 ***** using LISMOD language *****;
124 %macro BaseLISMOD;
125 lismod
126   yvar=com1-com6   disj1-disj6   proj1-proj6,
127   etavar=F_comm   F_dj   F_pj;
128 matrix _LAMBDAY_   [,1] =1 c2-c6 12*0,
129                   [,2] =6*0 1 d2-d6 6*0,
130                   [,3] =12*0 1 p2-p6;
131 matrix _THETAY_   [1,1]=vec1-vec6 ved1-ved6 vep1-vep6;
132 matrix _PSI_      [,] =VF1
133                   psi12 VF2
134                   psi13 psi23 VF3;
135 %mend;
136
137 /* multigroup: unconstrained model */
138 proc tcalis;
139 group 1 / data=group1 nobs=213;
140 group 2 / data=group2 nobs=213;
141 model 1 / group=1;
142   %BaseLISMOD
143 model 2 / group=2;
144   refmodel 1 / AllNewParms;
145 run;
146
147 /* multigroup: constrained model */
148 proc tcalis;
149 group 1 / data=group1 nobs=213;
150 group 2 / data=group2 nobs=213;
151 model 1 / group=1;
152   %BaseLISMOD
153 model 2 / group=2;
154   refmodel 1;
155 matrix _THETAY_   [1,1]=G2_vec1-G2_vec6 G2_ved1-G2_ved6 G2_vep1-G2_vep6;
156 matrix _PSI_      [,] =G2_VF1
157                   G2_psi12 G2_VF2
158                   G2_psi13 G2_psi23 G2_VF3;
159 run;

```

## IMPLEMENTING MULTIGROUP COMPARISONS USING THE PATH LANGUAGE

A PATH model can be specified by the PATH, PVAR, and PCOV (if needed) statements. First, a baseline model for each group is specified in lines 51 through 80 of Table 1, which is defined by a SAS macro. The name of this macro is BasePath. Defining a model by a SAS macro is not essential. The PATH model code can be inserted directly into a PROC TCALIS step to get the same results. Using a SAS macro can show the organization of the model specification more clearly.

The biggest advantage of the PATH language is that the path diagram can be translated easily into the PATH model. This conversion is very straightforward:

- Each single-headed arrow in the path diagram is specified in the PATH statement.
- Each double-headed arrow pointing to a single variable is specified in the PVAR statement.
- Each double-headed arrow pointing to two different variables is specified in the PCOV statement.

In the PATH statement, all loadings in the measurement model are translated to the corresponding path entries, and a fixed value or a unique parameter name is specified to each path entry. Specifying a parameter name to a path entry tells SAS that this particular parameter is freely estimated in the model. For example, the loading of F\_comm to the first indicator, com1, is fixed to 1, whereas the loading to the second indicator, com2, is freely estimated and the parameter name is c2. The direction of these path entries does not matter at all. A -> B is equivalent to B <- A.

In the PVAR statement, variance parameters for latent constructs and errors are specified with fixed values or unique parameter names. In this example, all variance parameters are freely estimated. Specifying a name followed by double underscores is a quick way to generate unique parameter names. The double underscores are replaced with

a unique number each time a new parameter name is generated. For example, the variances of F\_comm, F\_dj, and F\_pj, are specified as VF1, VF2, and VF3, respectively.

In the PCOV statement, the covariance parameters are specified: either error covariances or covariances between latent constructs. For example, the covariance between F\_comm and F\_dj is specified as psi12.

In lines 83 through 91, the established path model is fit to the two groups separately. Some commonly reported model fit indices for the baseline models are given in Table 2.

**Table 2.** Fit Indices for Baseline Models in Example

	Software Package	Chi-Square	df	SRMSR	CFI	NNFI	RMSEA	RMSEA (90% CI)
Group 1	SAS 9.2	322.12	132	.05	.95	.94	.082	(.071, .094)
	LISREL 8.80	322.12	132	.05	.97	.96	.075	(.063, .087)
	Mplus 5.21	323.64	132	.05	.95	.94	.083	(.071, .094)
Group 2	SAS 9.2	318.00	132	.06	.93	.91	.082	(.070, .093)
	LISREL 8.80	318.00	132	.06	.96	.96	.078	(.067, .090)
	Mplus 5.21	319.50	132	.06	.93	.91	.082	(.070, .093)

**Table 3.** SAS PROC TCALIS Model Fit Summary from the Unconstrained Model

The TCALIS Procedure Covariance Structure Analysis: Maximum Likelihood Estimation Fit Summary		
Modeling Info	N Observations	426
	N Variables	18
	N Moments	342
	N Parameters	78
	N Active Constraints	0
	Independence Model Chi-Square	6340.2212
Absolute Index	Independence Model Chi-Square DF	306
	Fit Function	1.5097
	Chi-Square	640.1148
	Chi-Square DF	264
	Pr > Chi-Square	<.0001
	Z-Test of Wilson & Hilferty	11.8665
Parsimony Index	Hoelter Critical N	202
	Root Mean Square Residual (RMSR)	0.1845
	Standardized RMSR (SRMSR)	0.0550
	Goodness of Fit Index (GFI)	0.8657
	Adjusted GFI (AGFI)	0.8260
	Parsimonious GFI	0.7469
Incremental Index	RMSEA Estimate	0.0820
	RMSEA Lower 90% Confidence Limit	0.0739
	RMSEA Upper 90% Confidence Limit	0.0901
	Probability of Close Fit	0.0000
	Akaike Information Criterion	112.1148
	Bozdogan CAIC	-1222.2572
	Schwarz Bayesian Criterion	-958.2572
	McDonald Centrality	0.6431
	Bentler Comparative Fit Index	0.9377
	Bentler-Bonett NFI	0.8990
	Bentler-Bonett Non-normed Index	0.9278
	Bollen Normed Index Rho1	0.8830
	Bollen Non-normed Index Delta2	0.9381
	James et al. Parsimonious NFI	0.7756

Second, an unconstrained model fit to the two groups simultaneously is estimated in lines 93 through 101. In the two GROUP statements, the data and sample size are explicitly specified for each group. It is worth noting that the current example has the same sample sizes across groups, but different sample sizes are allowed in general cases. In this specification, Group 1 uses the path model specified by invoking the BasePath macro, whereas Group 2 specifies the same model, but there is no cross-group constraint. Specifically, at line 100, the REFMODEL statement,

along with the AllNewParms option, within the scope of a MODEL statement, indicates that the pattern of fixed and free parameters in Model 2 is equivalent to Model 1, but the parameter names used in Model 2 are totally different. By default, SAS adds suffix, \_mdl2, to the parameter names in Model 1 to create new parameter names for Model 2.

It can be expected that parameter estimates of the unconstrained model are identical to the two separate baseline models just run, whereas the fit information is now combined as a single set of indices. In fact, some fit indices can be derived directly from the two baseline models, e.g., chi-square and degree of freedom. Table 3 gives the actual model fit summary for the unconstrained model, and Table 4 gives fit comparison among groups. Technical details of these fit indices calculated in PROC TCALIS can be found from the SAS/STAT user's guide (SAS Institute, 2008, pp.6892-6905).

**Table 4.** SAS PROC TCALIS Fit Comparison from the Unconstrained Model

The TCALIS Procedure				
Covariance Structure Analysis: Maximum Likelihood Estimation				
Fit Comparison Among Groups				
Modeling Info		Overall	Group 1	Group 2
	N Observations	426	213	213
	N Variables	18	18	18
	N Moments	342	171	171
	N Parameters	78	39	39
	N Active Constraints	0	0	0
	Independence Model Chi-Square	6340.2212	3689.4138	2650.8074
	Independence Model Chi-Square DF	306	153	153
Fit Index	Fit Function	1.5097	1.5194	1.5000
	Percent Contribution to Chi-Square	100	50	50
	Root Mean Square Residual (RMSR)	0.1845	0.1756	0.1930
	Standardized RMSR (SRMSR)	0.0550	0.0502	0.0594
	Goodness of Fit Index (GFI)	0.8657	0.8684	0.8630
	Bentler-Bonett NFI	0.8990	0.9127	0.8800

Third, a model with cross-group constraints placed on the corresponding factor loadings is established in lines 103 through 121. In this specification, Group 1 uses the path model specified by invoking the BasePath macro, whereas Group 2 specifies a new path model by integrating some replacements with the old path model. That is, the PVAR and PCOV statements in lines 111 through 120, nested within the scope of the REFMODEL statement in line 110, which is nested within the scope of the MODEL statement in line 109, replace the old specifications, and hence, new parameter names are specified for Model 2 variance and covariance parameters. Here, new parameter names are created by adding prefix G2\_ to the old parameter names. So far, the equality constraints are placed on the factor loadings by leaving the parameter names of factor loadings unchanged.

## IMPLEMENTING MULTIGROUP COMPARISONS USING THE LISMOD LANGUAGE

The same steps are applied in the LISMOD language, but the modeling philosophy is different. A LISMOD model is specified by the LISMOD and one or more MATRIX statements. First, a baseline model for each group is specified. In lines 124 through 135, a parallel SAS macro is defined to specify a LISMOD model. The name of this macro is BaseLISMOD. Like the original implementation of LISREL, the LISMOD language uses a matrix specification interface, which is characterized by two tasks. The first task is to define the variables in the model. The second task is to specify the parameters in the related matrices.

The first task is accomplished in the LISMOD statement. In lines 126 through 127, the YVAR= option specifies the y-side indicators, and the ETAVAR= option specifies y-side latent constructs. Special attention should be paid to the order of variables listed in those options because the order is implicitly used to define the variable order in rows and columns of the LISMOD model matrices.

The second task is accomplished by the MATRIX statements. In each statement, the model matrix is specified by using the matrix names defined in the LISMOD language. Fortunately, these matrix names are conceptually consistent with the LISREL matrix names. In this example, assuming all the measurement models are in the y-side, the \_LAMBDA\_ matrix is referred to as the loadings from the latent construct to the observed indicators, which is specified in line 128 through 130.

The starting location of a matrix and the different continuation direction for assigning fixed values or matrix element names are specified in the following different forms of elements in the bracket:

- The form of [i, j]/[i, j] specifies a vertically/horizontally continued matrix elements, starting at [i,1]/[1, j].



- The form of  $[i, j]$  specifies a diagonally continued matrix elements  $[i, j], [i+1, j+1], \dots, [i+n-1, j+n-1]$ .
- The form of  $[ , ]$  specifies all valid matrix elements starting at  $[1, 1]$  and continuing row-wise. For a symmetric matrix, valid matrix elements are those in the lower triangle.

The same steps to test multigroup invariance are followed as the previous PATH language section. Fit indices and parameter estimates using the LISMOD language are essentially the same as those by using the PATH language. Programming details in Table 1 are adequate for illustration purposes, and thus, discussion is omitted here.

## RESULTS

Before we compare the results from SAS to those from Mplus and LISREL, readers are encouraged to run the SAS code in Table 1 and look at the layout of the output generated by the different languages in PROC TCALIS. Not only are the languages similar to those from Mplus/LISREL, but the layout of the SAS output is remarkably similar to those from Mplus/LISREL as well.

### FIT INDICES

Compared to the values reported in Jones-Farmer et al. (2008), different results, except the degree of freedom, are obtained due to the different precision of the data. Table 2 reports some fit indices for the baseline model in Group 1 and Group 2. The same value of the chi-square index is obtained from SAS and LISREL, but Mplus reports a slightly higher value because the minimum fit function is multiplied by the sample size,  $n$ , instead of  $(n-1)$ . It is easily verified that, in Group 1, if we divide the chi-square value from SAS/LISREL by  $(n-1)$  and multiply it to  $n$ , we get  $(322.12/(312-1))*312 = 323.64$  in Mplus. In terms of the standardized root mean square residual (SRMSR), identical values are reported in SAS, LISREL, and Mplus. However, inconsistency occurs in CFI, NNFI, and RMSEA and in the 90% confidence interval of RMSEA. SAS essentially reaches quite good agreement with Mplus on these indices, but LISREL gives higher values for CFI and NNFI but lower values for RMSEA and the 90% confidence interval of RMSEA.

For the unconstrained and constrained models, a list of commonly used fit indices is reported in Table 5. First, the degree of freedom and the degree of freedom for the independence model reported in SAS, Mplus, and LISREL are identical (i.e., 264 and 279 for the unconstrained and constrained model separately, and 306 for the independence model) indicating that models are correctly specified.

**Table 5.** Fit Indices for the Constrained Model in Example

	Unconstrained Multigroup Model			Constrained Multigroup Model		
	SAS 9.2	LISREL 8.80	Mplus 5.21	SAS 9.2	LISREL 8.80	Mplus 5.21
Chi-square	640.11	640.11	643.13	677.16	677.16	680.36
df	264	264	264	279	279	279
Chi-square for Independence Model	6340.22	11533.99	6370.13	6340.22	11533.99	6370.13
df for Independence Model	306	306	306	306	306	306
Group 1 Standardized RMSR	.05	.05	N/A	.05	.05	N/A
Group 2 Standardized RMSR	.06	.06	N/A	.07	.07	N/A
Standardized RMSR	.06	N/A	.06	.06	N/A	.07
CFI	.94	.97	.94	.93	.96	.93
NNFI	.93	.96	.93	.93	.96	.93
RMSEA	.082	.077	.082	.082	.077	.082
RMSEA (90% CI)	(.074, .090)	(.068, .085)	(.074, .090)	(.074, .090)	(.069, .085)	(.074, .090)

Note. N/A means not reported by the software.

Second, in spite of the fact that the minimum fit function chi-squares are the same across SAS and LISREL, the chi-square for the independence model are drastically different between LISREL and SAS/Mplus. These chi-square fit indices are important in that many other fit indices depend on either the minimum fit function chi-square (e.g., RMSEA) or the chi-square for the independence model (e.g., NNFI and CFI). Readers interested in the detailed discussion of different types of chi-square statistics should refer to Jöreskog (2004).

Third, the SRMSR reported in SAS, Mplus, and LISREL are nearly identical. Table 4 shows that SAS has a complete report for individual groups and the overall model, whereas LISREL does not report the overall SRMSR, and Mplus does not report for individual groups.

Lastly, inconsistency remains on indices such as CFI, NNFI, RMSEA, and the 90% confidence interval of RMSEA. As before, SAS and Mplus agree well on values of CFI, NNFI, RMSEA, and the 90% confidence interval of RMSEA, but higher or lower values are reported from LISREL.



## PARAMETER ESTIMATES

Despite the various discrepancies in fit indices, parameter estimates are essentially equivalent regardless of the software packages among SAS, LISREL, and Mplus.

## CONCLUDING REMARKS

Historically, SEM models specified in SAS PROC CALIS are limited in scope. In this article, two types of modeling languages in the new PROC TCALIS, offering programming similarity and comfort to users of other major SEM software, are illustrated for multigroup model analysis. Comparing the fit indices and parameter estimates to those from Mplus and LISREL shows that PROC TCALIS is a valid procedure for SEM. Taken together, PROC TCALIS provides substantial assistance to multigroup analyses. Although some temporary issues are still being resolved, PROC TCALIS will eventually be rolled back to PROC CALIS in a future version of SAS. In terms of Monte Carlo simulation, where multigroup analysis is needed, it can be expected that the SAS system will be an ideal choice.

## REFERENCES

- Arbuckle, J. L., (2003). AMOS (Version 5.0) [Computer software]. Chicago: Smallwaters.
- Bentler, P. M., & Wu, E. J. C. (2002). EQS 6 for Windows guide. Encino, CA: Multivariate Software.
- Brown, T. A. (2006) Confirmatory factor analysis for applied research. New York, NY: Guilford Press.
- Byrne, B. M. (2004). Testing for multigroup invariance using AMOS graphics: A road less traveled. Structural Equation Modeling, 11(2), 272-300.
- Fan, X., & Fan, X. (2005). Using SAS for Monte Carlo simulation research in SEM. Structural Equation Modeling, 12(2), 299-333.
- Fan, X., Felsövályi, Á., Sivo, S. A., & Keenan, S. (2003). SAS for Monte Carlo studies: A guide for quantitative researchers. Cary, NC: SAS Institute.
- Fan, X., & Sivo, S. A., (2005). Sensitivity of fit indexes to misspecified structural or measurement model components: Rationale of two-index strategy revisited. Structural Equation Modeling, 12(3), 343-367.
- Jones-Farmer, L. A., Pitts J. P., & Rainer, R. K. (2008). A note on multigroup comparisons using SAS PROC CALIS. Structural Equation Modeling, 15, 154-173.
- Jöreskog, K. G. (2004). On chi-squares for the independence model and fit measures in LISREL. Retrieved March 1, 2010, from <http://www.ssicentral.com/lisrel/techdocs/ftb.pdf>.
- Jöreskog, K. G., & Sörbom, D. (1996). LISREL 8: User's reference guide. Chicago: Scientific Software International.
- Marcoulides, G. A., & Hershberger, S. L. (1997). Multivariate statistical methods: A first course. Mahwah, NJ: Lawrence Erlbaum Associates, Inc.
- Muthén, L. K., & Muthén, B. O. (1998-2007). Mplus user's guide. Los Angeles, CA: Muthén & Muthén.
- Neale, M.C., Boker, S. M., Xie, G., & Maes, H. H. (2003). Mx: statistical modeling. VCU Box 900126, Richmond, VA 23298: Department of Psychiatry. 6th Edition.
- SAS Institute Inc. (2008) SAS/STAT® 9.2 User's Guide. Cary, NC: SAS Institute Inc.
- Yang, Y., & Green, S. B., (2010). A note on structural equation modeling estimates of reliability. Structural Equation Modeling, 17(1), 66-81.

## ACKNOWLEDGMENTS

This work is supported by the University of Kansas' Life Span Institute/Research Design & Analysis unit (LSI/RDA) and the Center for Educational Testing & Evaluation (CETE).

## CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Fei Gu  
Psychology and Research in Education

University of Kansas  
1122 W Campus Rd.  
Lawrence, KS 66045  
785-864-9701  
fgu@ku.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.